FOURIER ANALYSIS – Programs in Mathematica

Nicolae Cotfas, version 07 May 2020 (for future updates see https://unibuc.ro/user/nicolae.cotfas/)

TRIGONOMETRIC **FOURIER SERIES** with period 2π

Extension by periodicity with period 2π of a function $f:[-\pi,\pi)\longrightarrow\mathbb{C}$ is $\varphi:\mathbb{R}\longrightarrow\mathbb{C},\quad \varphi(t)=f((t+\pi)\,(\text{modulo } 2\pi)-\pi)$

$$a_n\overset{\text{def}}{=}\frac{1}{\pi}\int_{-\pi}^{\pi}\varphi(t)\,\cos nt\,dt \qquad b_n\overset{\text{def}}{=}\frac{1}{\pi}\int_{-\pi}^{\pi}\varphi(t)\,\sin nt\,dt$$

$$S(t)=\frac{1}{2}a_0+\sum_{n=1}^{\infty}(a_n\cos nt+b_n\sin nt)$$

$$\varphi(t)\approx S_k(t)=\frac{1}{2}a_0+\sum_{n=1}^{k}(a_n\cos nt+b_n\sin nt)$$

```
In[1]=  k:=3;    f[t_]:=t^2/7+Cos[t]-1/2;    phi[t_]:=f[Mod[t+Pi,2 Pi]-Pi]
        a[n_]:=(1/Pi) Integrate[phi[t] Cos[n t],{t,-Pi,Pi}]
        b[n_]:=(1/Pi) Integrate[phi[t] Sin[n t],{t,-Pi,Pi}]
        Sk[t_]=a[0]/2+Sum[a[n] Cos[n t]+b[n] Sin[n t],{n,1,k}]
        Plot[Mod[t+Pi,2 Pi]-Pi, {t, -3 Pi, 3 Pi}, AspectRatio -> 0.2]
        Plot[phi[t],{t,-3 Pi,3 Pi}, AspectRatio -> 0.3]
        Plot[Sk[t], {t,-3 Pi,3 Pi}, AspectRatio -> 0.3]
        Show[{%,%%}]
```



TRIGONOMETRIC **FOURIER SERIES** with period T

Extension by periodicity with period T of a function $f:[0,T)\longrightarrow\mathbb{C}$ is $\varphi:\mathbb{R}\longrightarrow\mathbb{C},\quad \varphi(t)=f(t\,(\text{modulo } T))$

$$a_n\overset{\text{def}}{=}\frac{2}{T}\int_0^T\varphi(t)\,\cos\frac{2\pi}{T}nt\,dt \qquad b_n\overset{\text{def}}{=}\frac{2}{T}\int_0^T\varphi(t)\,\sin\frac{2\pi}{T}nt\,dt$$

$$S(t)=\frac{1}{2}a_0+\sum_{n=1}^{\infty}\left(a_n\cos\frac{2\pi}{T}nt+b_n\sin\frac{2\pi}{T}nt\right)$$

$$\varphi(t)\approx S_k(t)=\frac{1}{2}a_0+\sum_{n=1}^{k}\left(a_n\cos\frac{2\pi}{T}nt+b_n\sin\frac{2\pi}{T}nt\right)$$

```
In[1]=  k:=2;    f[t_]:=Sin[t];    T:=5;    phi[t_]:=f[Mod[t,T]]
        a[n_]:=(2/T) Integrate[phi[t] Cos[2 Pi n t/T],{t,0,T}]
        b[n_]:=(2/T) Integrate[phi[t] Sin[2 Pi n t/T],{t,0,T}]
        Sk[t_]=a[0]/2+Sum[a[n] Cos[2 Pi n t/T]+b[n] Sin[2 Pi n t/T],{n,1,k}]
        Plot[Mod[t,T], {t,-T, 2 T}, AspectRatio -> 0.3]
        Plot[phi[t],{t,-T,2 T}, AspectRatio -> 0.3]
        Plot[Sk[t], {t,-T, 2 T}, AspectRatio -> 0.3]
        Show[{%,%%}]
```



FOURIER SERIES with period 2π

Extension by periodicity with period 2π of a function $f:[-\pi,\pi)\longrightarrow\mathbb{C}$ is $\varphi:\mathbb{R}\longrightarrow\mathbb{C},\quad \varphi(t)=f((t+\pi)\,(\text{modulo }2\pi)-\pi)$

$$c_n\overset{\text{def}}{=}\frac{1}{2\pi}\int_{-\pi}^{\pi}\varphi(t)\,\mathrm{e}^{-int}\,dt \qquad S(t)=\sum_{n=-\infty}^{\infty}c_n\,\mathrm{e}^{int} \qquad \varphi(t)\approx S_k(t)=\sum_{n=-k}^{k}c_n\,\mathrm{e}^{int}$$

```
In[1]=  k:=5;     f[t_]:=-t;         phi[t_]:=f[Mod[t+Pi,2 Pi]-Pi]
        c[n_]:=(1/(2 Pi)) Integrate[Exp[-I n t] f[t],{t,-Pi,Pi}]
        Sk[t_]=Sum[c[n] Exp[I n t],{n,-k,k}]
        Plot[Mod[t+Pi,2 Pi]-Pi,{t,-3 Pi,3 Pi}, AspectRatio -> 0.2]
        Plot[phi[t],{t,-3 Pi,3 Pi}, AspectRatio -> 0.3]
        Plot[Sk[t],{t,-3 Pi,3 Pi}, AspectRatio -> 0.3]
        Show[{%,%%}]
```






FOURIER SERIES with period T

Extension by periodicity with period $T=b-a$ of an arbitrary function $f:[a,b)\longrightarrow\mathbb{C}$ is $\varphi:\mathbb{R}\longrightarrow\mathbb{C},\quad \varphi(t)=f((t-a)\,(\text{modulo }T)+a)$

$$c_n\overset{\text{def}}{=}\frac{1}{T}\int_{a}^{b}\varphi(t)\,\mathrm{e}^{-\frac{2\pi\mathrm{i}}{T}nt}\,dt \qquad S(t)=\sum_{n=-\infty}^{\infty}c_n\,\mathrm{e}^{\frac{2\pi\mathrm{i}}{T}nt} \qquad \varphi(t)\approx S_k(t)=\sum_{n=-k}^{k}c_n\,\mathrm{e}^{\frac{2\pi\mathrm{i}}{T}nt}$$

```
In[1]=  k:=5;     f[t_]:=t^2;     a:=-1;     b:=2;     T:=b-a;     phi[t_]:=f[Mod[t-a,T]+a]
        c[n_]:=(1/T) Integrate[Exp[-2 Pi I n t/T] f[t],{t,a,b}]
        Sk[t_]=Sum[c[n] Exp[2 Pi  I n t/T],{n,-k,k}]
        Plot[Mod[t-a,T]+a,{t,a-T,b+T}, AspectRatio -> 0.2]
        Plot[phi[t],{t,a-T,b+T}, AspectRatio -> 0.3]
        Plot[Sk[t],{t,a-T,b+T}, AspectRatio -> 0.3]
        Show[{%,%%}]
```






DISCRETE (FINITE) **FOURIER TRANSFORM**

Extension by periodicity of a function

Extension by periodicity with period No of the function $f:\{0,1,\ldots,No-1)\longrightarrow\mathbb{C},\quad f(n)=n$	is	$\psi:\mathbb{Z}\longrightarrow\mathbb{C},\quad \psi(n)=n\ (\text{modulo } No)$
Extension by periodicity with period $No=n_1-n_0+1$ of the function $f:\{n_0,n_0+1,\ldots,n_1\}\longrightarrow\mathbb{C},\quad f(n)=n$	is	$\psi:\mathbb{Z}\longrightarrow\mathbb{C},\quad \psi(n)=(n-n_0)\ (\text{modulo } No)+n_0$
Extension by periodicity with period $No=n_1-n_0+1$ of an arbitrary function $f:\{n_0,n_0+1,\ldots,n_1\}\longrightarrow\mathbb{C}$	is	$\varphi:\mathbb{Z}\longrightarrow\mathbb{C},\quad \varphi(n)=f(\psi(n))$ where $\psi(n)=(n-n_0)\ (\text{modulo } No)+n_0$

```
In[1]=  f[n_]:=n^2/100;  n0:=-10;  n1:=10;  No:=n1-n0+1;  psi[n_]:=Mod[n-n0,No]+n0;  phi[n_]:=f[psi[n]]
        ListPlot[Table[{n,Mod[n,No]},{n,n0-No,n1+No}],Filling->Axis,PlotStyle->PointSize[Medium],
                                                                  PlotRange->All,AspectRatio->0.3]
        ListPlot[Table[{n,psi[n]},{n,n0-No,n1+No}],Filling->Axis,PlotStyle->PointSize[Medium],
                                                                  PlotRange->All,AspectRatio->0.3]
        ListPlot[Table[{n,phi[n]},{n,n0-No,n1+No}],Filling->Axis,PlotStyle->PointSize[Medium],
                                                                  PlotRange->All,AspectRatio->0.3]
```







■ Approximation of a periodic function $\varphi:\mathbb{Z}\longrightarrow\mathbb{C}$ by using only a part of the values of its Fourier transform

Remark. If you use an older version of Mathematica, write `Cos[t]+I Sin[t]` instead of `Exp[t]`.

Notations and formulas

$f:\{n_0,n_0+1,...,n_1\}\longrightarrow\mathbb{C}$
$No=n_1-n_0+1$
$\psi:\mathbb{Z}\longrightarrow\mathbb{C},\quad \psi(n)=(n-n_0)\ (\text{modulo } No)+n_0$
$\varphi:\mathbb{Z}\longrightarrow\mathbb{C},\quad \varphi(n)=f(\psi(n))$
psi $\equiv \psi$
phi $\equiv \varphi$
Fphi $\equiv F[\varphi]$
phiapprox $\equiv \varphi_{approx}$

$$F[\varphi](k)\overset{\text{def}}{=}\sum_{k=n_0}^{n_1}\mathrm{e}^{-\frac{2\pi\mathrm{i}}{No}kn}\varphi(n)$$

$$\varphi_{approx}(n)=\frac{1}{No}\sum_{k=-m}^{m}\mathrm{e}^{\frac{2\pi\mathrm{i}}{No}nk}F[\varphi](k)$$

■
```
In[1]=  m:=2;  f[n_]:=n^2/100;  n0:=-10;  n1:=10;  No:=n1-n0+1;  psi[n_]:=Mod[n-n0,No]+n0;  phi[n_]:=f[psi[n]]
        Fphi[k_] := Sum[Exp[-2 Pi I k n/No] phi[n], {n, n0, n1}]
        phiapprox[n_,m_]=(1/No) Sum[Exp[2 Pi I k n/No]  Fphi[k],{k, -m, m}]
        ListPlot[Table[{n, phi[n]},{n,n0-No,n1+No}],Filling->Axis,PlotStyle->PointSize[Medium],
                                                                PlotRange->All,AspectRatio->0.3]
        ListPlot[Table[{n, phiapprox[n,m]},{n,n0-No,n1+No}],Filling->Axis,PlotStyle->PointSize[Medium],
                                                                PlotRange->All,AspectRatio->0.3]
```





■
```
In[1]=  m:=4;  f[n_]:=n^2/100;  n0:=-8;  n1:=12;  No:=n1-n0+1;  psi[n_]:=Mod[n-n0,No]+n0;  phi[n_]:=f[psi[n]]
        Fphi[k_] := Sum[Exp[-2 Pi I k n/No] phi[n], {n, n0, n1}]
        phiapprox[n_,m_]=(1/No) Sum[Exp[2 Pi I k n/No]  Fphi[k],{k, -m, m}]
        ListPlot[Table[{n, phi[n]},{n,n0-No,n1+No}],Filling->Axis,PlotStyle->PointSize[Medium],
                                                                PlotRange->All,AspectRatio->0.3]
        ListPlot[Table[{n, phiapprox[n,m]},{n,n0-No,n1+No}],Filling->Axis,PlotStyle->PointSize[Medium],
                                                                PlotRange->All,AspectRatio->0.3]
```





■
```
In[1]=  m:=4;  f[n_]:=n^2/100;  n0:=-5;  n1:=15;  No:=n1-n0+1;  psi[n_]:=Mod[n-n0,No]+n0;  phi[n_]:=f[psi[n]]
        Fphi[k_] := Sum[Exp[-2 Pi I k n/No] phi[n], {n, n0, n1}]
        phiapprox[n_,m_]=(1/No) Sum[Exp[2 Pi I k n/No]  Fphi[k],{k, -m, m}]
        ListPlot[Table[{n, phi[n]},{n,n0-No,n1+No}],Filling->Axis,PlotStyle->PointSize[Medium],
                                                                PlotRange->All,AspectRatio->0.3]
        ListPlot[Table[{n, phiapprox[n,m]},{n,n0-No,n1+No}],Filling->Axis,PlotStyle->PointSize[Medium],
                                                                PlotRange->All,AspectRatio->0.3]
```





Extension by periodicity of a function

Formulas and notations

$$\boxed{\begin{array}{l} f:\{n_0,n_0+1,\dots,n_1\}\times\{m_0,m_0+1,\dots,m_1\}\longrightarrow\mathbb{C}, \\ \qquad\qquad\qquad N_x=n_1-n_0+1, \qquad N_y=m_1-m_0+1, \\ \varphi:\mathbb{Z}\times\mathbb{Z}\longrightarrow\mathbb{C}, \qquad \varphi(n,m)=f((n-n_0)\,(\text{modulo }N_x)+n_0,\,(m-m_0)\,(\text{modulo }N_y)+m_0) \end{array}}$$

phi $\equiv \varphi$
n0 $\equiv n_0$
n1 $\equiv n_1$
m0 $\equiv m_0$
m1 $\equiv m_1$
Nx $\equiv N_x$
Ny $\equiv N_y$

```
In[1]=  n0:=-3; n1:=3; m0:=-4; m1=4; f[n_, m_]:=N[Exp[-(n^2+m^2)/4]]
        Nx:=n1-n0+1; Ny:=m1-m0+1; phi[n_, m_]:=f[Mod[n-n0,Nx]+n0,Mod[m-m0,Ny]+m0]
        DiscretePlot3D[f[n,m], {n,n0,n1},{m,m0,m1}, ExtentSize ->0.7,  AxesLabel -> Automatic, AspectRatio ->1.5]
        DiscretePlot3D[phi[n,m],{n,n0-Nx,n1+Nx},{m,m0-Ny,m1+Ny}, ExtentSize -> 0.7, AxesLabel -> Automatic, AspectRatio -> 0.7]
        Image[Table[f[n,m],{n,n0,n1},{m,m0,m1}]]
        Image[Table[phi[n,m],{n,n0-Nx,n1+Nx},{m,m0-Ny, m1+Ny}]]
```









■ Function reconstruction by using only a part of the values of its Fourier transform

Formulas and notations

$$\boxed{\begin{array}{l}\varphi:\{n_0,n_0+1,\dots,n_1\}\times\{m_0,m_0+1,\dots,m_1\}\longrightarrow\mathbb{C},\\ \quad N_x=n_1-n_0+1,\qquad N_y=m_1-m_0+1,\end{array}}$$

$$\boxed{\begin{array}{l}F[\varphi]:\mathbb{Z}\times\mathbb{Z}\longrightarrow\mathbb{C},\qquad F[\varphi](k,\ell)=F[\varphi](k+N_x,\ell+N_y)\\ \qquad F[\varphi](k,\ell)\overset{\text{def}}{=}\sum_{n=n_0}^{n_1}\sum_{m=m_0}^{m_1}\mathrm{e}^{-\frac{2\pi\mathrm{i}}{N_x}k\,n}\,\mathrm{e}^{-\frac{2\pi\mathrm{i}}{N_y}\ell\,m}\,\varphi(n,m)\end{array}}$$

$$\boxed{F^{-1}[\psi](k,\ell)\overset{\text{def}}{=}\frac{1}{N_x\,N_y}\sum_{n=0}^{N_x-1}\sum_{m=0}^{N_y-1}\mathrm{e}^{\frac{2\pi\mathrm{i}}{N_x}k\,n}\,\mathrm{e}^{\frac{2\pi\mathrm{i}}{N_y}\ell\,m}\,\psi(n,m)}$$

$$\boxed{\varphi=F^{-1}[F[\varphi]]}\quad\Longrightarrow\quad\boxed{\varphi(n,m)=\frac{1}{N_xN_y}\sum_{k=0}^{N_x-1}\sum_{\ell=0}^{N_y-1}\mathrm{e}^{\frac{2\pi\mathrm{i}}{N_x}k\,n}\,\mathrm{e}^{\frac{2\pi\mathrm{i}}{N_y}\ell\,m}\,F[\varphi](k,\ell)}$$

$$\boxed{\varphi_{approx}(n,m)=\frac{1}{N_x\,N_y}\sum_{k=-a}^{a}\sum_{\ell=-b}^{b}\mathrm{e}^{\frac{2\pi\mathrm{i}}{N_x}k\,n}\,\mathrm{e}^{\frac{2\pi\mathrm{i}}{N_y}\ell\,m}\,F[\varphi](k,\ell)}$$

`n0` $\equiv n_0$
`n1` $\equiv n_1$
`m0` $\equiv m_0$
`m1` $\equiv m_1$
`Nx` $\equiv N_x$
`Ny` $\equiv N_y$
`phi` $\equiv \varphi$
`Fphi` $\equiv F[\varphi]$
`phiapprox` $\equiv \varphi_{approx}$

```
In[1]=  a:=2; b:=2; n0:=-3; n1:=3; m0:=-4; m1 =4; Nx:=n1-n0+1; Ny:=m1-m0+1; phi[n_, m_] := N[Abs[Sin[(n + m)/5]]]
        Fphi[k_, l_] :=N[Sum[Exp[2 Pi I k n/Nx] Exp[2 Pi I l m/Ny] phi[n,m], {n,n0,n1}, {m,m0, m1}]]
        phiapprox[n_,m_]:=N[Re[(1/(Nx Ny))Sum[ Exp[-2 Pi I k n/Nx]Exp[-2 Pi I l m/Ny] Fphi[k,l], {k,-a,a},{l,-b,b}]]]
        DiscretePlot3D[phi[n,m],{n,n0,n1},{m,m0,m1},ExtentSize -> 0.7, AxesLabel -> Automatic, AspectRatio -> 0.7]
        DiscretePlot3D[phiapprox[n,m],{n,n0,n1},{m,m0,m1}, ExtentSize -> 0.7, AxesLabel -> Automatic, AspectRatio -> 0.7]
        ImageRotate[Image[Table[phi[n, m], {n, n0, n1}, {m, m0, m1}]],Right]
        ImageRotate[Image[Table[phiapprox[n, m], {n, n0, n1}, {m, m0, m1}]],Right]
```









■ Fourier transform in Image processing

■
```
In[1]=  n0=-5; n1=5; m0=-5; m1=5
        f1[n_,m_]:=Mod[Sin[n m],1]
        f2[n_,m_]:=Mod[(n/10)^2 - (m/10)^2,1]
        f3[n_,m_]:=Mod[n m/100,1]
        ImageRotate[Image[Table[{f1[n,m]},{n,n0,n1}, {m, m0,m1}]],Left]
        ImageRotate[Image[Table[{f1[n,m],0,0},{n,n0,n1}, {m, m0,m1}]],Left]
        Image[Table[{f1[n,m],f2[n,m],f3[n,m]}, {n,n0,n1}, {m,m0,m1}]]
```

  

■
```
In[1]=  foto := Import["C:\\Users\\Apus-de-soare.jpg"]
        Show[foto]
        ImageRotate[Image[Table[PixelValue[foto, {n,k}], {n,100,120}, {k,180,260}]], Left]
        MatrixForm[Table[PixelValue[foto, {n,k}], {n,100,103}, {k,180,183}]]
        MatrixForm[Table[PixelValue[foto, {n,k}][[1]], {n,100,103}, {k,180,183}]]
        MatrixForm[Table[PixelValue[foto, {n,k}][[2]], {n,100,103}, {k,180,183}]]
        MatrixForm[Table[PixelValue[foto, {n,k}][[3]], {n,100,103}, {k,180,183}]]
```

 

$$\begin{pmatrix} \begin{pmatrix}0.121569\\0.0941176\\0.\end{pmatrix} & \begin{pmatrix}0.137255\\0.105882\\0.0156863\end{pmatrix} & \begin{pmatrix}0.14902\\0.105882\\0.027451\end{pmatrix} & \begin{pmatrix}0.133333\\0.0862745\\0.\end{pmatrix} \\ \begin{pmatrix}0.129412\\0.0901961\\0.\end{pmatrix} & \begin{pmatrix}0.137255\\0.105882\\0.0156863\end{pmatrix} & \begin{pmatrix}0.152941\\0.109804\\0.0235294\end{pmatrix} & \begin{pmatrix}0.137255\\0.0901961\\0.00392157\end{pmatrix} \\ \begin{pmatrix}0.133333\\0.0941176\\0.\end{pmatrix} & \begin{pmatrix}0.14902\\0.109804\\0.0117647\end{pmatrix} & \begin{pmatrix}0.156863\\0.113725\\0.027451\end{pmatrix} & \begin{pmatrix}0.137255\\0.0901961\\0.00392157\end{pmatrix} \\ \begin{pmatrix}0.133333\\0.0941176\\0.\end{pmatrix} & \begin{pmatrix}0.14902\\0.109804\\0.0117647\end{pmatrix} & \begin{pmatrix}0.156863\\0.113725\\0.027451\end{pmatrix} & \begin{pmatrix}0.14902\\0.0901961\\0.\end{pmatrix} \end{pmatrix}$$

```
( 0.12156862745098039`  0.13725490196078433`  0.14901960784313725`  0.13333333333333333` )
( 0.12941176470588237`  0.13725490196078433`  0.15294117647058825`  0.13725490196078433` )
( 0.13333333333333333`  0.14901960784313725`  0.1568627450980392`   0.13725490196078433` )
( 0.13333333333333333`  0.14901960784313725`  0.1568627450980392`   0.14901960784313725` )
```

```
( 0.09411764705882353`  0.10588235294117647`  0.10588235294117647`  0.08627450980392157` )
( 0.09019607843137255`  0.10588235294117647`  0.10980392156862745`  0.09019607843137255` )
( 0.09411764705882353`  0.10980392156862745`  0.11372549019607843`  0.09019607843137255` )
( 0.09411764705882353`  0.10980392156862745`  0.11372549019607843`  0.09019607843137255` )
```

```
( 0.`  0.01568627450980392`   0.027450980392156862`  0.`                    )
( 0.`  0.01568627450980392`   0.023529411764705882`  0.00392156862745098`   )
( 0.`  0.011764705882352941`  0.027450980392156862`  0.00392156862745098`   )
( 0.`  0.011764705882352941`  0.027450980392156862`  0.`                    )
```

Fourier transform of functions

Original function $\boxed{\varphi:\mathbb{R}\longrightarrow\mathbb{C}}$ $\mapsto$ Fourier transform $\boxed{F[\varphi]:\mathbb{R}\longrightarrow\mathbb{C},\quad F[\varphi](p) \overset{\text{def}}{=} \int_{-\infty}^{\infty} e^{ipx}\,\varphi(x)\,dx}$

```
In[1]=  phi[x_] := If[Abs[x] <= 0.5, 1, 0]
        Fphi[p_] := Integrate[Exp[I p x] phi[x], {x, -Infinity, Infinity}]
        Plot[phi[x], {x, -3, 3}]
        Plot[Fphi[p], {p, -30, 30}, PlotRange -> All]
```






```
In[1]=  phi[x_] := If[Abs[x] <= 1, 1-Abs[x], 0]
        Fphi[p_] := Integrate[Exp[I p x] phi[x], {x, -Infinity, Infinity}]
        Plot[phi[x], {x, -3, 3}]
        Plot[Fphi[p], {p, -30, 30}, PlotRange -> All]
```






```
In[1]=  phi[x_] := Exp[-x^2]
        Fphi[p_] := Integrate[Exp[I p x] phi[x], {x, -Infinity, Infinity}]
        Plot[phi[x], {x, -3, 3}]
        Plot[Fphi[p], {p, -30, 30}, PlotRange -> All]
```





■ Description of the difraction

$$\underset{\boxed{\varphi:\mathbb{R}\longrightarrow[0,1]}}{\text{Screen transparency}} \quad\mapsto\quad \underset{\boxed{|F[\varphi]|^2:\mathbb{R}\longrightarrow[0,\infty)}}{\text{Diffraction pattern}}$$

```
In[1]=  phi[x_] := If[Abs[x] <= 1, 1, 0]
        ParametricPlot[{phi[x], x}, {x, -3, 3}]
        Plot[phi[x], {x, -3, 3}]
        ParametricPlot[{Abs[FourierTransform[phi[x],x,p]]^2,p}, {p,-10,10}, PlotRange -> All, Exclusions -> None, AspectRatio -> 0.5]
```






```
In[1]=  phi[x_] := If[Abs[x-1] <=0.2, 1, 0]+If[Abs[x+1] <=0.2, 1, 0]
        ParametricPlot[{phi[x], x}, {x, -3, 3}]
        Plot[phi[x], {x, -3, 3}]
        ParametricPlot[{Abs[FourierTransform[phi[x],x,p]]^2,p}, {p,-20,20}, PlotRange -> All, Exclusions -> None, AspectRatio -> 0.5]
```






```
In[1]=  phi[x_] :=If[Abs[x - 1.5] <= 0.5, 1 - 2 Abs[x - 1.5], 0] +If[Abs[x + 1.5] <= 0.5, (2 Abs[x + 1.5])^2, 0]
        ParametricPlot[{phi[x], x}, {x, -3, 3}]
        Plot[phi[x], {x, -3, 3}]
        ParametricPlot[{Abs[FourierTransform[phi[x],x,p]]^2,p}, {p,-15,15}, PlotRange -> All, Exclusions -> None, AspectRatio -> 0.5]
```